// File: contracts/Lock.sol



//          █████  ███████ ███████ ███████ ████████ ██      ██ ███    ██ ██   ██
//         ██   ██ ██      ██      ██         ██    ██      ██ ████   ██ ██  ██
//         ███████ ███████ ███████ █████      ██    ██      ██ ██ ██  ██ █████
//         ██   ██      ██      ██ ██         ██    ██      ██ ██  ██ ██ ██  ██
//         ██   ██ ███████ ███████ ███████    ██    ███████ ██ ██   ████ ██   ██
//


// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

/**
 * @title AssetLink Team & Advisors Tokens Lock
 * @version 1.0.2
 * @date 2023-08-01
 * @license MIT
 * @author Tech Department, AssetLink
 *
 * @dev Smart contract to lock AssetLink's team and advisors tokens.
 * This contract locks the team and advisors tokens for a period of 2 years,
 * with a subsequent quarterly release of 25% of the remaining tokens.
 * This mechanism is designed to align the team's interests with those of the investors,
 * ensuring a long-term commitment to the project's success.
 *
 * Key Features:
 * - 2 years initial lock.
 * - 25% release every 3 months after the initial lock period.
 * - Only the designated beneficiary can release and claim the tokens.
 * - Built-in checks for release conditions.
 **/

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "@openzeppelin/contracts/security/ReentrancyGuard.sol";

interface IMainContract is IERC20 {
    function handleReleasedTokens(uint256 amount) external;
}

contract ASETTeamTokensLock is ReentrancyGuard {
    address public beneficiary;
    uint256 public initialReleaseTime;
    uint256 public lastReleaseTime;
    IERC20 public token;
    uint256 public releasedPercentage = 0;
    bool public initialized = false;

    event Initialized(address indexed token);
    event TokensReleased(uint256 amount);

    // Modifiers
    modifier onlyBeneficiary() {
        require(
            msg.sender == beneficiary,
            "Only the beneficiary can call this function"
        );
        _;
    }

    modifier onlyOnce() {
        require(!initialized, "Already initialized");
        _;
    }

    constructor() {
        beneficiary = msg.sender;
        initialReleaseTime = block.timestamp + 730 days; // 2 years from now
    }

    /**
     * @dev Initializes the contract with the token address.
     * @param _token Address of the token contract.
     */
    function initialize(address _token) external onlyBeneficiary onlyOnce {
        require(
            _token != address(0),
            "Token address cannot be the zero address"
        );
        token = IERC20(_token);
        initialized = true;
        emit Initialized(_token);
    }

    /**
     * @dev Releases a portion of the locked tokens to the beneficiary.
     */
    function release() public onlyBeneficiary nonReentrant {
        require(initialized, "Contract not initialized");
        require(
            block.timestamp >= initialReleaseTime,
            "Tokens are still locked"
        );
        require(
            releasedPercentage < 100,
            "All tokens have already been released"
        );

        uint256 monthsSinceLastRelease = (block.timestamp - lastReleaseTime) / 30 days;
        require(
            monthsSinceLastRelease >= 3 || releasedPercentage == 0,
            "Less than 3 months since last release or tokens already released this quarter"
        );

        uint256 amountToRelease = (token.balanceOf(address(this)) * 25) / 100; // 25% of remaining balance
        releasedPercentage += 25;

        lastReleaseTime = block.timestamp;

        require(amountToRelease > 0, "No tokens to release");

        // Transfer tokens to the beneficiary
        token.transfer(beneficiary, amountToRelease);

        emit TokensReleased(amountToRelease);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}